応用複素関数レポート課題3

桂田 祐史

2024年7月8日,2024年7月9日

目 次

1 課題3 1

2 〆切, 提出の仕方, 参考情報など 1

3 注意 (必ず読むこと) 3

4 FAQ (よくされる質問) 3

5 サンプルプログラム potential2d-v-2024.edp 解説 4

6 流線を描くために: 流れ関数 ψ を求める 5

A サンプル・プログラム解説 5

A.1 一様流のプログラム sample0.edp 5

A.2 sample1.edp, sample2.edp 7

1 課題3

2次元非圧縮ポテンシャル流の定常流で、流体の占める領域 Ω と、その境界 $\Gamma = \partial\Omega$ での流速の法線成分 $V_n := \boldsymbol{v}\cdot\boldsymbol{n}$ が分かっている場合に、速度ポテンシャル ϕ を計算して、等ポテンシャル線と速度場を可視化せよ。

領域 Ω としては、円盤領域以外のものを選ぶこと。境界値 (流速の法線成分) V_n も自分で興味のあるもの、自分の都合の良いものを選んで良い (後の注意を読んでおくこと)。

以下でサンプル・プログラムを紹介する。それを理解した上で、領域 Ω と境界データ V_n を自分が決めたものに書き換えれば良い。自由度は高いので、工夫・遊び心発揮を期待する。

もし出来れば、流線も可視化せよ。流線の書き方には色々なやり方があるが、流れ関数を求めるやり方を勧めておく (それを求めるにもポテンシャル問題が使えたりするので)。

2 〆切, 提出の仕方, 参考情報など

- 〆切は7月31日 (水曜) 22:00です (注: 定期試験は 7/29(月) に終了する予定)。
- 提出方法: Oh-o! Meiji に A4 サイズの PDF で提出して下さい。
  容量制限 (1ファイル 30MB) に引っかかった場合は、複数のファイルに分割するなど工夫して下さい。

- 使用するプログラミング言語は、自分の MacBook で実行して見せることが可能なものであればなんでも可。
  (と言っていますが、実際上は FreeFem++ となるでしょう。FreeFem++ に関する質問は気軽にして下さい。)

- プログラムとその実行結果、実行するための情報を含めること。プログラムは別ファイルにしても良いです (こちらが調べやすいので)。

- FreeFem++ の使い方については、

  - `FreeFEM-documentation.pdf` (公式ドキュメント)
    https://doc.freefem.org/pdf/FreeFEM-documentation.pdf
    (本当は Mac の中にあるのですが、FreeFem++ のバージョンによって場所が違うので…[1])
  - 「FreeFem++ の紹介」桂田が書いた紹介文書
    https://m-katsurada.sakura.ne.jp/labo/text/welcome-to-freefem/
  - 「FreeFem++ノート」桂田が書いた文法メモ
    https://m-katsurada.sakura.ne.jp/labo/text/freefem-note/

- サンプル・プログラムを用意しました。入手法をまとめて示しておきます。

```
curl -O https://m-katsurada.sakura.ne.jp/program/fem/poisson-kikuchi.edp
curl -O https://m-katsurada.sakura.ne.jp/complex2/potential2d-v-2024.edp
curl -O https://m-katsurada.sakura.ne.jp/complex2/sample0.edp
curl -O https://m-katsurada.sakura.ne.jp/complex2/sample1.edp
curl -O https://m-katsurada.sakura.ne.jp/complex2/sample2.edp
```

  `poisson-kikuchi.edp` は、正方形領域における Poisson 方程式の境界値問題を解くプログラムで、すでに紹介済みです。これについては、「FreeFem++ で菊地 [1] の Poisson 方程式の例題を解く」を見て下さい。多角形領域の扱い方が分かるはずです。

  それ以外の `potential2d-v-2024.edp`, `sample0.edp`, `sample1.edp`, `sample2.edp` は、円盤領域で

$$\text{(1a)} \qquad \triangle\phi = 0 \quad (\text{in } \Omega)$$

$$\text{(1b)} \qquad \frac{\partial\phi}{\partial \boldsymbol{n}} = V_n \quad (\text{on } \partial\Omega)$$

  を解くプログラムです。中身はほとんど同じですが、境界条件の設定の仕方が違います。

弱形式は、すでに説明したように

$$\iint_\Omega (\phi_x v_x + \phi_y v_y)\, dx\, dy = \int_{\partial\Omega} V_n v\, d\sigma \quad (\text{任意の試験関数 } v \text{ に対して}).$$

この左辺は FreeFem++ のプログラム中では、`int2d(Th)(dx(phi)*dx(v)+dy(phi)*dy(v))` のようにすれば良いですが、右辺は少し工夫が必要です (FreeFem++ の機能が今ひとつなためです)。

[1] 一応 `find /Applications/FreeFem++.app -name FreeFEM-documentation.pdf -print` で表示されますが…

3　注意 (必ず読むこと)

次の (1) が非常に重要である。例年それを間違えてナンセンスなレポートを提出する人が少なくない。

(1) V_n は $\int_{\partial\Omega} V_n \, d\sigma = 0$ を満たしている必要がある。実際、$\frac{\partial\phi}{\partial n}$ の定義と Gauss の発散定理から

$$\int_{\partial\Omega} V_n \, d\sigma = \int_{\partial\Omega} \frac{\partial\phi}{\partial n} \, d\sigma = \int_{\partial\Omega} \operatorname{grad}\phi\cdot\boldsymbol{n} \, d\sigma = \int_{\Omega} \operatorname{div}\operatorname{grad}\phi \, d\boldsymbol{x} = \int_{\Omega} \triangle\phi \, d\boldsymbol{x} = \int_{\Omega} 0 \, d\boldsymbol{x} = 0.$$

非圧縮流 ($\operatorname{div}\boldsymbol{v} = 0$) の速度場 $\boldsymbol{v}$ が分かっている場合に、$V_n = \boldsymbol{v}\cdot\boldsymbol{n}$ と置いた場合は、この条件は必ず成り立っている ($\int_{\partial\Omega} V_n \, d\sigma = \int_{\partial\Omega} \boldsymbol{v}\cdot\boldsymbol{n} \, d\sigma = \int_{\Omega} \operatorname{div}\boldsymbol{v} \, d\boldsymbol{x} = \int_{\Omega} 0 \, d\boldsymbol{x} = 0$)。

そうでない場合には、$\int_{\partial\Omega} V_n \, d\sigma = 0$ が成り立つように、注意深く V_n を選ぶ必要がある。

過去の例では、円を楕円にするような (一見) 安直な選択をしたレポートが多かった、$\boldsymbol{n}$ が意外と簡単ではないので (おおぜい間違えました)、注意すること。

(2) 湧き出しや吸い込み、点渦など、特異点が Ω 内にあるような問題 (レポート課題 2 のテーマであったとも言える) は、この方法では解くことが出来ない。

4　FAQ (よくされる質問)

「境界値 V_n を場合分けを含む関数としたいが、FreeFem++ がプログラムを受け付けてくれません。どうすればいいですか？」

(FreeFem++ ってしょうがないですね…) 例えば以下の 2 つの解決策があります。

(a) 弱形式には複数の `int1d()` が指定できるので、境界を分割して、その部分ごとに (場合分けを含まない) 境界値を与えるようにプログラムを書く。

(b) 例えば `(x>1 && y>0)` のような式は、条件が成り立つならば 1, 成り立たないならば 0 という値を持つので、`if` を使わずに、式だけで場合分けを含む関数が記述できる。

次のサンプル・プログラムでは、(b) を用いていますが、ある程度以上複雑になった場合は、私のおすすめは (a) です。付録の `sample1.edp` を見て下さい。

5 サンプルプログラム potential2d-v-2024.edp 解説

potential2d-v-2024.edp

```
1  // potential2d-v-2024.edp
2  //   curl -O https://m-katsurada.sakura.ne.jp/complex2/potential2d-v-2024.edp
3  //   2 次元非圧縮ポテンシャル流
4  //     速度ポテンシャル，速度を求め、等ポテンシャル線，速度場を描く
5
6  border Gamma(t=0,2*pi) { x = cos(t); y = sin(t); } // 円盤領域
7  int m=40;
8  mesh Th=buildmesh(Gamma(m));
9  plot(Th, wait=1, ps="Th.eps");
10 // 次の 2 行は区分 1 次多項式を使うという意味
11 fespace Vh(Th,P1);
12 Vh phi, v;
13 // 境界条件の設定 右上と左下のみ Vn=x+2y（これは確かに ∫ Vn d σ=0 を満たす）
14 func Vn=((x>0&&y>0) || (x<0&&y<0))*(x+2*y);
15
16 // 速度ポテンシャルφを求める
17 solve Laplace(phi,v) =
18   int2d(Th)(dx(phi)*dx(v)+dy(phi)*dy(v)) -int1d(Th,Gamma)(Vn*v);
19
20 // 平均を 0 にする（解の一意性がないため、時々生じる大きなズレを消す
21 real meanphi = int2d(Th)(phi)/Th.area; // Th.area は int2d(Th)(1); でも計算可能
22 phi = phi - meanphi;
23
24 // φの等高線（等ポテンシャル線）を描く
25 plot(phi,ps="contourpotential.eps",wait=1);
26
27 // ベクトル場 (v1,v2)=∇φ を描く（ちょっと雑なやり方）
28 Vh v1, v2;
29 v1=dx(phi); v2=dy(phi);
30 plot([v1,v2],ps="vectorfield.eps",wait=1);
31
32 // 等ポテンシャル線とベクトル場を同時に描く
33 plot([v1,v2],phi,ps="both.eps", wait=1);
```

- 6 行目で領域 Ω の境界 $\Gamma := \partial\Omega$ を指定している (それで Ω が決まる)。
- 8 行目で、Γ を m 分割して、Γ の囲む範囲を三角形分割して、それを mesh 型の変数 Th に代入している。
- 11 行目、有限要素空間 V_h を区分的 1 次関数の空間と定義している。この辺についてはこの講義では説明を省略する (知りたい人は有限要素法のテキストを読んで下さい)。
- 12 行目、ϕ と試験関数 v を Vh の要素とする。
- 14 行目、境界値 V_n の設定をしている。ここでは

$$V_n(x,y) = \begin{cases} x+2y & ((x>0 \text{ かつ } y>0) \text{ または } (x<0 \text{ かつ } y<0) \text{ のとき}) \\ 0 & (\text{それ以外}) \end{cases}$$

としている。(後の例 sample0.edp では $V_n(x,y) = x+2y = \begin{pmatrix} 1 \\ 2 \end{pmatrix} \cdot \boldsymbol{n}$ としていて、これは $\boldsymbol{v} = \begin{pmatrix} 1 \\ 2 \end{pmatrix}$ という一様流となる。) これは確かに $\displaystyle\int_{\partial\Omega} V_n \, d\sigma = 0$ を満たしている。

- 17〜18 行目、弱形式の定義。

- 21〜22行目、Neumann境界値問題は定数だけの不定性を持つため (ϕ が解ならば ϕ+定数も解)、時々大きなズレが生じ、解の描画などで問題が生じる。平均値 $\dfrac{\iint_\Omega \phi(x,y)dx\,dy}{\iint_\Omega dx\,dy}$ を引くことで、平均 $=0$ にしておく。

6　流線を描くために: 流れ関数 ψ を求める

これは少し難しいので、レポートの必要条件とはしない。しかし流線があると流れの様子がよく分かるので、描く価値は高い。以下は頑張ろうという人向けのヒントである。講義で

$$\psi(\boldsymbol{x}) = \int_{C_{\boldsymbol{x}}} \psi_x\,dx + \psi_y\,dy = \int_{C_{\boldsymbol{x}}} \begin{pmatrix} -v \\ u \end{pmatrix} \cdot d\boldsymbol{r} = \int_{C_{\boldsymbol{x}}} \begin{pmatrix} -v \\ u \end{pmatrix} \cdot \boldsymbol{t}\,ds \quad (\boldsymbol{x} \in \overline{\Omega})$$

という式を紹介した ($C_{\boldsymbol{x}}$ は定点 $\boldsymbol{a}$ と $\boldsymbol{x}$ を結ぶ曲線)。領域が Jordan 領域 (1つの単純閉曲線で囲まれた領域) であれば、$\boldsymbol{x}$ が境界上の点である場合、定点 $\boldsymbol{a}$ と $C_{\boldsymbol{x}}$ を境界上に選ぶことで、($\begin{pmatrix} -v \\ u \end{pmatrix} \cdot \boldsymbol{t} = \boldsymbol{v} \cdot \boldsymbol{n}$. また境界上での $\boldsymbol{v} \cdot \boldsymbol{n} = V_n$ は知っていると仮定しているので)

$$\psi(\boldsymbol{x}) = \int_{C_{\boldsymbol{x}}} \boldsymbol{v} \cdot \boldsymbol{n}\,ds = \int_{C_{\boldsymbol{x}}} V_n\,ds \quad (\boldsymbol{x} \in \partial\Omega).$$

これから、境界上の点 $\boldsymbol{x}$ における ψ の値 $g(\boldsymbol{x}) := \psi(\boldsymbol{x})$ $(x \in \partial\Omega)$ が得られる。それを用いて

$$\triangle\psi = 0 \quad (\text{in } \Omega), \quad \psi(\boldsymbol{x}) = g(\boldsymbol{x}) \quad (\text{on } \partial\Omega)$$

という Laplace 方程式の Dirichlet 境界値問題を解くことで流れ関数 ψ が得られる。

A　サンプル・プログラム解説

以下すべて単位円版領域において

$$\triangle\phi = 0 \quad \text{in } \Omega, \quad \frac{\partial\phi}{\partial n} = V_n \quad \text{on } \partial\Omega$$

を解くプログラムである。境界条件の指定の部分が違っている。

A.1　一様流のプログラム `sample0.edp`

サンプルプログラム `sample0.edp` では、単位円盤領域 $\Omega = \{(x,y) \in \mathbb{R}^2 \mid x^2+y^2 < 1\}$ における一様流 $\boldsymbol{v} = \begin{pmatrix} 1 \\ 2 \end{pmatrix}$ の場合のプログラムである。単位円盤なので $\boldsymbol{n} = \begin{pmatrix} x \\ y \end{pmatrix}$ (ここは良く考えること。例えば楕円の場合は $\boldsymbol{n}$ はかなり複雑な式になる。). これから

$$V_n = \boldsymbol{v} \cdot \boldsymbol{n} = \begin{pmatrix} 1 \\ 2 \end{pmatrix} \cdot \begin{pmatrix} x \\ y \end{pmatrix} = x + 2y.$$

プログラムでは、`func Vn=x+2*y;` で `Vn` を定義して、弱形式中で `int1d(Th,Gamma)(Vn*v)` と使っている。

($\boldsymbol{v}$ が Ω で定数関数なので) $\operatorname{div} \boldsymbol{v} = 0$ であるから、当然 $\int_\Gamma V_n \, d\sigma = 0$ も成り立つ。

sample0.edp

```
// sample0.edp
//   https://m-katsurada.sakura.ne.jp/complex2/sample0.edp
//   2 次元非圧縮ポテンシャル流
//     速度ポテンシャル，速度を求め、等ポテンシャル線，速度場を描く

border Gamma(t=0,2*pi) { x = cos(t); y = sin(t); } // 円盤領域
int m=60;
mesh Th=buildmesh(Gamma(m));
plot(Th, wait=1, ps="Th.eps");
// 次の 2 行は区分 1 次多項式を使うという意味
fespace Vh(Th,P1);
Vh phi, v, v1, v2;
// 境界条件の設定
func Vn=x+2*y;// Ωが単位円で，V=(1,2) のとき V・n=x+2y

// 速度ポテンシャルφを求め、その等高線（等ポテンシャル線）を描く
solve Laplace(phi,v) =
  int2d(Th)(dx(phi)*dx(v)+dy(phi)*dy(v)) -int1d(Th,Gamma)(Vn*v);

real meanphi=int2d(Th)(phi)/Th.area;
phi=phi-meanphi;
plot(phi,ps="contourpotential.eps",wait=1);

// ベクトル場 (v1,v2)=∇φ を描く（ちょっと雑なやり方）
v1=dx(phi); v2=dy(phi);
plot([v1,v2],ps="vectorfield.eps",wait=1);

// 等ポテンシャル線とベクトル場を同時に描く
plot([v1,v2],phi,ps="both.eps", wait=1);
```



図 1: 一様流 (sample0.edp)

A.2 sample1.edp, sample2.edp

単位円盤領域の境界である単位円周上で、$0 \le \theta \le \pi/6$, $\pi/6 \le \theta \le \pi$, $\pi \le \theta \le 4\pi/3$ $4\pi/3 \le \theta \le 2\pi$ の範囲をそれぞれ $\Gamma_1, \Gamma_2, \Gamma_3, \Gamma_4$ とする。すなわち

$$\begin{aligned}
\Gamma_1 &= \{(\cos\theta, \sin\theta) \mid 0 \le \theta \le \pi/6\}, \\
\Gamma_2 &= \{(\cos\theta, \sin\theta) \mid \pi/6 \le \theta \le \pi\}, \\
\Gamma_3 &= \{(\cos\theta, \sin\theta) \mid \pi \le \theta \le 4\pi/3\}, \\
\Gamma_4 &= \{(\cos\theta, \sin\theta) \mid 4\pi/3 \le \theta \le 2\pi\}.
\end{aligned}$$

境界条件 $\frac{\partial\phi}{\partial n} = V_n$ における V_n は、次のようになっているとする。

$$V_n = \begin{cases} -2 & (\text{on } \Gamma_1) \\ 0 & (\text{on } \Gamma_2) \\ 1 & (\text{on } \Gamma_3) \\ 0 & (\text{on } \Gamma_4). \end{cases}$$

念のため $\int_{\partial\Omega} V_n \, d\sigma = 0$ の確認

$\int_\gamma d\sigma = \gamma$ の長さ に注意すると

$$\begin{aligned}
\int_{\partial\Omega} V_n \, d\sigma &= \int_{\Gamma_1} V_n \, d\sigma + \int_{\Gamma_2} V_n \, d\sigma + \int_{\Gamma_3} V_n \, d\sigma + \int_{\Gamma_4} V_n \, d\sigma \\
&= -2 \int_{\Gamma_1} d\sigma + 0 + 1 \cdot \int_{\Gamma_3} d\sigma + 0 = -2 \cdot \frac{\pi}{6} + 1 \cdot \frac{\pi}{3} = 0.
\end{aligned}$$

sample1.edp, sample2.edp はともにこの問題を解くプログラムである。

もしも、プログラム中で V_n がうまく定義できれば、sample0.edp と同様に

```
solve Laplace(phi,v) =
   int2d(Th)(dx(phi)*dx(v)+dy(phi)*dy(v)) -int1d(Th,Gamma)(Vn*v);
```

というコードが使える。

しかし V_n を 1 行の関数として定義するのはあまり簡単でない。一応出来なくはなくて、それは後で紹介するが (sample2.edp)、ここでは次のようにすることを勧める。

要点は

$$\int_{\partial\Omega} V_n v \, d\sigma = \sum_{j=1}^{4} \int_{\Gamma_j} V_n v \, d\sigma = \int_{\Gamma_1} V_n v \, d\sigma + \int_{\Gamma_3} V_n v \, d\sigma$$

と分解することである (Γ_2, Γ_4 では $V_n = 0$ であるから、$\int_{\Gamma_j} V_n v \, d\sigma$ $(j = 2, 4)$ は必要ない)。

Γ_1, Γ_3 における V_n を、それぞれ Vn1, Vn3 と定義すれば

```
solve Laplace(phi,v) =
   int2d(Th)(dx(phi)*dx(v)+dy(phi)*dy(v))
  -int1d(Th,Gamma1)(Vn1*v)-int1d(Th,Gamma3)(Vn3*v);
```

とすれば良い 。このためには、もちろんプログラム中で $\Gamma_1, \Gamma_2, \Gamma_3, \Gamma_4$ を用意する必要がある

が、それは難しくない (境界を 4 つに分けた正方形領域のプログラム poisson-kikuchi.edp と同様のことをすれば良い)。以上の考察から、次のプログラムが得られる。

sample1.edp

```
// sample1.edp
//   https://m-katsurada.sakura.ne.jp/complex2/sample1.edp
//   2 次元非圧縮ポテンシャル流
//     速度ポテンシャル，速度を求め、等ポテンシャル線，速度場を描く

border Gamma1(t=0,pi/6)      { x = cos(t); y = sin(t); }
border Gamma2(t=pi/6,pi)     { x = cos(t); y = sin(t); }
border Gamma3(t=pi,4*pi/3)   { x = cos(t); y = sin(t); }
border Gamma4(t=4*pi/3,2*pi) { x = cos(t); y = sin(t); }
int m=5;
mesh Th=buildmesh(Gamma1(m)+Gamma2(5*m)+Gamma3(2*m)+Gamma4(4*m));
plot(Th, wait=1, ps="Th.eps");
// 次の 2 行は区分 1 次多項式を使うという意味
fespace Vh(Th,P1);
Vh phi, v;

// Gamma1, Gamma3 上での Vn
func Vn1 = -2.0;
func Vn3 = 1.0;

// 速度ポテンシャルφを求め、その等高線（等ポテンシャル線）を描く
solve Laplace(phi,v) =
  int2d(Th)(dx(phi)*dx(v)+dy(phi)*dy(v))
  -int1d(Th,Gamma1)(Vn1*v)-int1d(Th,Gamma3)(Vn3*v);

real mean=int2d(Th)(phi)/Th.area; // Th.area は int2d(Th)(1.0)
phi=phi-mean;
plot(phi,ps="contourpotential.eps",wait=1);

// ベクトル場 (v1,v2)=∇φ を描く（ちょっと雑なやり方）
Vh v1, v2;
v1=dx(phi); v2=dy(phi);
plot([v1,v2],ps="vectorfield.eps",wait=1);

// 等ポテンシャル線とベクトル場を同時に描く
plot([v1,v2],phi,ps="both.eps", wait=1);
```



図 2: sample1.edp の結果 ($\theta \in [0, \pi/6]$ で勢い良く入り、$\theta \in [\pi, 4\pi/3]$ でゆっくり出て行く)

以下、参考までに、V_n を 1 命令 (長いので 2 行になっている) で書いたプログラム sample2.edp を紹介しておく。いわゆる偏角の主値 (値が $(-\pi, \pi]$ に属する) を返す関数 atan2(y,x) を使うが、やや不自然な感じがするのは否めない (角度の範囲が $[0, 2\pi)$ でないことにも注意が必要である)。

強引に 1 行関数を作る

```
func Vn=((atan2(y,x)>=0.0&&atan2(y,x)<=pi/6) * (-2.0)
       + (atan2(y,x)>=-pi&&atan2(y,x)<=-2.0*pi/3.0) * (1.0));
```

sample2.edp

```
// sample2.edp
//   https://m-katsurada.sakura.ne.jp/complex2/sample2.edp
//   2 次元非圧縮ポテンシャル流
//     速度ポテンシャル，速度を求め、等ポテンシャル線，速度場を描く

border Gamma(t=0,2*pi) { x = cos(t); y = sin(t); } // 円盤領域
int m=60;
mesh Th=buildmesh(Gamma(m));
plot(Th, wait=1, ps="Th.eps");
// 次の 2 行は区分 1 次多項式を使うという意味
fespace Vh(Th,P1);
Vh phi, v, v1, v2;
// 境界条件の設定 (atan2(y,x) は [-pi,pi) の範囲の値を返す)
func Vn=((atan2(y,x)>=0.0&&atan2(y,x)<=pi/6) * (-2.0)
       + (atan2(y,x)>=-pi&&atan2(y,x)<=-2.0*pi/3.0) * (1.0));

// 速度ポテンシャルφを求め、その等高線 (等ポテンシャル線) を描く
solve Laplace(phi,v) =
  int2d(Th)(dx(phi)*dx(v)+dy(phi)*dy(v)) -int1d(Th,Gamma)(Vn*v);
real meanphi=int2d(Th)(phi)/Th.area;
phi=phi-meanphi;
plot(phi,ps="contourpotential.eps",wait=1);

// ベクトル場 (v1,v2)=∇φ を描く (ちょっと雑なやり方)
v1=dx(phi); v2=dy(phi);
plot([v1,v2],ps="vectorfield.eps",wait=1);

// 等ポテンシャル線とベクトル場を同時に描く
plot([v1,v2],phi,ps="both.eps", wait=1);
```



図 3: sample2.edp の結果 ($\theta \in [0, \pi/6]$ で勢い良く入り、$\theta \in [\pi, 4\pi/3]$ でゆっくり出て行く)